A Computer Routine for Calculating Total Lake Volume Contents of a Dissolved Substance from an Arbitrary Distribution of Concentration Profiles

F. M. Boyce

Environment Canada
Environnement Canada

A Computer Routine for Calculating Total Lake Volume Contents of a Dissolved Substance from an Arbitrary Distribution of Concentration Profiles

A Method of Calculating Lakewide Contents of Dissolved Substances

F. M. Boyce

TECHNICAL BULLETIN NO. 83

INLAND WATERS DIRECTORATE,
CANADA CENTRE FOR INLAND WATERS,
BURLINGTON, ONTARIO, 1973.

Cat. No.: En 36-503/83

CONTRACT #02KXKL327-3-8060
THORN PRESS LIMITED

Contents

Illustrations

A Computer Routine for Calculating Total Lake Volume Contents of a Dissolved Substance from an Arbitrary Distribution of Concentration Profiles

A Method of Calculating Lakewide Contents of Dissolved Substances

F. M. Boyce

INTRODUCTION

In support of our studies of Great Lake climatology and energy balance at the Canada Centre for Inland Waters, we have sought to develop methods for the computation of lake total stored heat based on BT temperature profiles collected during quasi-synoptic survey cruises. The method described here is simple in concept, yet general in application. Any arbitrary distribution of profiles within the lake may be used. The effects of bottom topography are taken into account. The method can be used to compute the total quantity of any dissolved substance in a lake. Adaptations of the technique have been developed to compute lake volume above and/or below any surface defined by an arbitrary distribution of measured depths of that surface (e.g. epilimnion and hypolimnion volumes).

PROGRAM SPLØTCH – A PROGRAM TO COMPUTE TOTAL LAKE VOLUME CONTENTS OF A PARAMETER FROM AN ARBITRARY DISTRIBUTION OF CONCENTRATION PROFILES

Let x_k, y_k, $k = 1, 2, \ldots, K$ be the horizontal coordinates of K sampling points within a lake. At each sampling point the concentration profile of a parameter is determined as a function of depth $C_k(Z)$, $0 \leq Z \leq Z_k$, where Z_k is the maximum depth to which the sampling is extended at the k^{th} station.

The profiles are not, in general, determined synoptically, and since there may be considerable variation of the instantaneous profiles about "mean" profiles characteristic of the period under investigation, we conclude that efforts designed to construct smooth surfaces of constant concentration by interpolation among groups of adjoining profiles are largely wasted. We propose to estimate the lake content of the substance by what amounts to a weighted sum of the individual vertically-integrated concentration profiles.

$$C_T = \sum_{\ell=1}^{L-1} \sum_{k=1}^{K} \int_{Z_\ell}^{Z_{\ell+1}} A_{\ell k}(Z)\, C_k(Z)\, dz \qquad (1)$$

The summation is by profile and by layer where the lake is divided vertically into L-1 layers by the surfaces. $Z - Z_\ell = 0$; $\ell = 1, 2, \ldots, L$.

This division is necessary because the horizontal areas of influence assigned to each station depend on the horizontal distances separating adjoining stations, and the profile data may not extend to the bottom, or if they do, the station position may not be the deepest point within the area assigned to that station. For a given layer then, we take into consideration only those profiles which extend through the layer. In other words

$$A_{\ell k}(Z) = 0 \text{ when } Z_k < Z_{\ell+1} \qquad (2)$$

$A_{\ell k}(Z_\ell)$ represents the area of the surface $Z = Z_\ell$ assigned to station k, and $A_{\ell k}(Z)$, $Z_\ell \leq Z \leq Z_{\ell+1}$ is the vertical projection of $A_{\ell k}(Z_\ell)$ on the horizontal plane at level Z, exclusive of that portion which may lie within the bottom of the lake. Figure 1 illustrates this definition. $A_{\ell k}(Z_\ell)$ is defined according to the horizontal distribution of the stations whose profiles extend at least to depth $Z_{\ell+1}$ and $A_{\ell k}(Z)$, $Z_\ell \leq Z \leq Z_{\ell+1}$ depends on the bottom topography of the lake as well.

In a given layer, the exclusion of profiles which penetrate deeply into the layer but which do not pierce through it may appear to waste valuable data. On the other hand, the adding of extra logic to an already complicated program in order to get around this difficulty, and imperfectly at that, does not seem worthwhile. In the present version of this program we rely on a judicious choice of layer depths to ensure adequate accuracy. The ultimate procedure would be to define the layer sequence according to the sequence of maximum sampling depths at the cost of a greatly-increased computing time.

One method frequently employed to define the individual areas of influence of a distribution of measurement points on a horizontal plane is to assign to each point those portions of the plane which lie closer to it than to any

other point. This is the Thiessen polygon approach for which a simple graphical procedure may be devised (Thiessen, 1911). Similar areas of influence may be generated by computer in the following manner.



Figure 1. Diagram defining the function $A_{\ell k}(Z)$.

1. A grid of square, equal-area elements is superposed on the lake surface, and ordered by the indices I, J I = 1, ..., IMAX, J = 1, ... , JMAX. The mesh size δ is determined by the accuracy required and by the memory available in the computer. The shoreline of the lake fits within the "rectangle" defined by the lines x = 0, x = IMAXδ, y = 0, y = JMAXδ.

2. The bottom topography of the lake is represented in the memory of the computer by the array ZBØ (I,J) I = 1, 2, ..., IMAX, J = 1, 2, ..., JMAX where ZBØ (I,J) is the average depth of the lake over the element or cell of index I, J. Those cells which lie outside the boundaries of the lake are assigned the value 0 in the array ZBØ (I,J).

3. A second array NØ (I,J) is stored in the computer and ultimately serves as a map of the areas of influence assigned to each station. Each station is assigned an identifying number $NSTAT_k$ and the area of influence of the k^{th} station is defined as those cells for which NØ(I,J) has the value $NSTAT_k$, once NØ(I,J) is filled. The filling of the array (NØ(I,J)) for the ℓ^{th} layer (which extends from Z = Z_ℓ to Z = $Z_{\ell+1}$) is described in steps 4 to 7.

4. NØ(I,J) is initialized by assigning the "no stations" index 999 to all cells whose depth is less than or equal to Z_ℓ.

5. A selection is made of the stations for which the profile information extends to depth Z = $Z_{\ell+1}$ or deeper, and the indices of the cells containing those station points, I_{0k}, J_{0k} are computed from the geographical position of the station. NØ(I_{0k}, J_{0k}) is assigned the value $NSTAT_k$ (see Appendix A for details).

6. An algorithm defines sequences of cells of relative index* I_m, J_m m = 1, 2 . . . . . whose centres lie within the annulus defined by the radii mδ and (m-1)δ. A simple weighting factor is assigned to each cell whose centre lies in the annulus, the factor being proportional to the radial distance of the centre of the cell from the circle of radius mδ (see Appendix B). For each step in the above sequence (m value) the elements NØ(I_{0k} + Im, J_{0k} + Jm) are assigned the value $NSTAT_k$ unless previously filled with another station number or the "no station" value. The weighting factor is used to decide which area of influence has the most "right" to a particular cell. As m increases, the area of influence of each station expands concentrically about the central point, with the demarcation between adjacent stations closely approximating to the perpendicular bisector of the lines joining them. The process continues until all the elements of NØ(I,J) have been assigned a non-zero value.

We now proceed to the computation of lake volume content of the parameter in steps 7 through 9.

7. The function $A_{\ell k}(Z)$ for each station is computed by sweeping the arrays NØ(I,J) and ZBØ(I,J) and counting

(a) the number of cells for which NØ(I,J) = $NSTAT_k$ (proportional to $A_{\ell k}(Z_\ell)$), $N_{\ell k}$

(b) the number of cells for which NØ(I,J) = $NSTAT_k$ and ZBØ(I,J) > $Z_{\ell+1}$ (proportional to $A_{\ell k}(Z_{\ell+1})$), $M_{\ell k}$

(c) the combined total vertical length of all the cells in count (a) between the planes Z = Z_ℓ, Z = $Z_{\ell+1}$ (proportional to the volume of the portion of the ℓ^{th} layer assigned to station k) = $V_{\ell k}$.

The constant of proportionality is δ^2, the surface area of a unit cell. The function $A_{\ell k}(Z)$ is approximated by the second degree polynomial

$$A_{\ell k}(Z) = Y_{1\ell k} + Y_{2\ell k}Z + Y_{3\ell k}Z^2 \qquad (3)$$

where the constants $Y_{p\ell k}$, p = 1,2,3 are defined by requiring that

*Relative to a central cell of arbitrary index.

$$A_{\ell k}(Z_\ell) = \delta^2 N_{\ell k}$$

$$A_{\ell k}(Z_{\ell+1}) = \delta^2 M_{\ell k} \qquad (4)$$

$$\int_{Z_\ell}^{Z_{\ell+1}} A_{\ell k}(z)dz = \delta^2 V_{\ell k}$$

8. The remainder of the computation for the k^{th} layer consists of forming the sum

$$\sum_{k=1}^{K} \int_{Z_\ell}^{Z_{\ell+1}} A_{\ell k}(z)\, C_k(z)dz \qquad (5)$$

$C_k(Z)$ is represented as a sequence of C, Z pairs (C_{qk}, Z_{qk}) and a profile is constructed by linear interpolation between adjacent pairs (see Appendix C). The integrals in (5) can be expressed in a simple closed form.

9. The steps 1 through 8 are repeated for each layer forming the sum (1). The bottom of the last layer may be taken as the maximum sampling depth, or the integration may be terminated at an arbitrary preselected depth. There is no *a priori* restriction on the number of layers.

SPLØTCH has been written and tested on temperature data collected in Lake Ontario. A listing is included as Appendix D.

CONCLUSIONS

Program SPLØTCH and its derivatives have been in use for several months at the Canada Centre. The method is particularly useful in studies where repeated surveys are made of the same lakes. The program adjusts to changes in the sample pattern and all computations are referred to a standard lake volume as defined by the digital representation of the lake. Changes in lake level are easily incorporated (see Appendix D). We have assembled digital maps on a 2 km grid of Lakes Ontario, Erie and Huron, and of Lake Okanagan in British Columbia. Versions of this program are available for CDC 6400, CDC 6600, machines and for CDC 3300 installations with disc storage.

ACKNOWLEDGEMENTS

The help of Miss Betty Pyde in programming the original version for a CDC 6400 computer is gratefully acknowledged. Mr. Ed Brunton has adapted this routine to a CDC 3300 machine.

REFERENCES

BOYCE, F.M. 1971. Proposals for the International Field Year on the Great Lakes. The Heat Storage of Lake Ontario. Canada Centre for Inland Waters Internal Report.

THIESSEN, A.H. 1911. Precipitation Averages for Large Areas. Monthly Weather Rev. July 1911. p. 1082.

APPENDIX A

CONSTRUCTION OF DIGITAL MAPS OF LAKE BATHYMETRY AND TRANSFORMATION FROM GEOGRAPHICAL COORDINATES TO MAP COORDINATES, I.J.

A hydrographic chart of the lake drawn on a polyconic projection is used. A system of rectangular coordinates x, y is chosen so that the axes align themselves with the sides of the rectangle of minimum area which surrounds the region to be graded (Figure A1). A square grid of mesh length δ is drawn on this area. Each cell is labelled by the indices (I, J), I denoting the cell number along the x-axis from the origin (generally eastwards) and J, the cell number from the origin along the y-axis (generally northwards). The point (x,y) in rectangular coordinates falls into the cell (I,J) where

$$I = \text{integral part of } \frac{x}{\delta} + 1$$
$$J = \text{integral part of } \frac{y}{\delta} + 1 \qquad \text{(A.1)}$$

To convert from geographical coordinates, latitude and longitude, to map coordinates x and y, we first define

$$g = Gm - G$$
$$p = \varphi - \varphi m \qquad \text{(A.2)}$$

where (φm, Gm) are the geographical coordinates of the map origin (latitude and longitude) (x=y=0) and (φ,G) are the geographical coordinates of the point. For maps of a sufficiently small area, x and y are adequately expressed in terms of p and g by the quadratic expressions

$$x = a_1 g + a_2 p + a_3 pg + a_4 g^2 + a_5 p^2$$
$$y = b_1 g + b_2 p + b_3 pg + b_4 g^2 + b_5 p^2$$

The coefficients a_i, b_i (i=1,2, . . .,5) may be computed from the theory of chart projections but it is probably simpler to determine them empirically once the grid has been drawn on the chart.



$$x_{max} = IMAX\ \delta + \frac{\delta}{2}$$
$$y_{max} = JMAX\ \delta + \frac{\delta}{2}$$

Figure A1. Definition of coordinate system.

The bathymetry of the lake is represented by the array ZBØ (I,J) I=1, IMAX, J=1, JMAX. Each element of the array represents the mean depth of the cell of index I,J. Cells lying outside the lake are assigned zero depth. The array is coded and punched on 80-column cards.

APPENDIX B

DESCRIPTION OF THE ALGORITHMS FOR ASSIGNING AREAS OF INFLUENCE TO AN ARBITRARY COLLECTION OF STATIONS

Starting with a central cell of index (I_0, J_0), the algorithm finds the indices of cells relative to this point (I_w, J_w) whose centres fall within the annulus of radii $(M-1)\delta$ and $M\delta$, where M is an integer and δ is the mesh length of the grid. For each (I_w, J_w) a weighting factor W is determined.

$$W = 0.1 \frac{M^2 - I_w^2 - J_w^2}{2M - 1}$$

This factor ranges monotonically from 0 to 0.1 as the centre of the cell moves from the outer to the inner limits of the annulus. Figure B1 is a flow chart of the position of the algorithm which determines I_w, J_w and w in the first octant.

The subroutine FILL computes first the other relative indices I_r, J_r = 1, 8 using the symmetry properties of the circle. Then for each station pertinent to the calculation (k



Figure B1. Algorithm generating indices of cells whose centres lie within the annulus of radii (M-1)DLAT and (M)DLAT with respect to a central cell.



Figure B2. Flow chart of the cell-filling algorithm.

= 1, K) whose position is defined by the central indices I_0 (k), J_0 (k), the bookkeeping array NØ (I,J) is assigned the values

$$N\emptyset[I_0(k) + I_r, J_0(k) + J_r]$$

(i) = k + W if the previous value was zero (unassigned).

(ii) Left unaltered if it had been assigned a value on the previous step of the M-sequence.

(iii) If a value had been assigned on a previous step of the k-sequence, the nonintegral part of the previous value is compared with W. If this nonintegral part is less than W, the new value of NØ ($I_0(k) + I_r$, $J_0(k) + J_r$) becomes k + W. Otherwise NØ is left unaltered.

Case (i) corresponds to an unassigned cell which is assigned provisionally to station k as the "first comer". Case (ii) occurs when the cell in question is closer to the originally assigned station. Case (iii) resolves the fine differences that arise at a given step of the radial sequence (M-sequence). Otherwise the distribution of cells would depend on the ordering of the stations.

Figure B2 is a flow chart of subroutine FILL.

The M-sequence is continued until the entire array NØ(I,J) is filled, that is to say, every cell of the lake below the upper depth of the layer in question is assigned to the station nearest it.

APPENDIX C

DATA FORMAT

For each station whose profile data pass through a given layer, it is required to form the integral

$$C_{\ell k} = \int_{Z_\ell}^{Z_{\ell+1}} A_{\ell k}(Z) C_k^Z(Z) dZ \quad \begin{pmatrix} (\ell^{th} \text{ layer}) \\ (k^{th} \text{ station}) \end{pmatrix} \quad (C.1)$$

The central program, it will be recalled, expresses $A_{\ell k}(Z)$ in the form

$$A_{\ell k}(Z) = Y_{1\ell k} + Y_{2\ell k} Z + Y_{3\ell k} Z^2 \quad (C.2)$$

Although there is no theoretical restriction on the form of the concentration profile, $C_k(Z)$, we have found it convenient to express the profile as a series of linear segments for which the integral (C.1) can be expressed in closed form (an algebraic sum of a 2p polynomials of 4th order in Z where p is the number of segments contained in the layer $Z_\ell \leq Z \leq Z_{\ell+1}$ (Boyce, 1971).

In the case where the data consist of concentrations measured at a finite number of depths, this approach amounts to a linear interpolation of concentration between pairs of data points. In the case of a continuous analog plot of concentration with depth such as is produced by a mechanical or electronic bathythermograph, we have found that a representation of the continuous profile by up to 8 linear segments gives results which are consistent with the



STN.#	YR	MO	DAY	HOUR	LAT °+'	LONG °+'	SFC.T	T_1	D_1	T_2	D_2	T_3	D_3	T_4	D_4	T_5	D_5	T_6	D_6	T_7	D_7	T_8	D_8	D_8			CONS #
203	70	10	05	191	43489	78410	124	124	003	089	006	084	007	081	010	069	016	066	021	042	026	041	051	052			71

DIGITIZED ABSTRACT FOR PUNCH CARD ENTRIES
RECORD FOR: LIMNOS 70-0-35
1908z
70/10/05
STN. B

Figure C1. Example of digitized temperature profile.

experimental accuracy of the measurements. That is to say, the difference between the integrals

$$\int_{Z_\ell}^{Z_{\ell+1}} C_k(Z)dZ - \int_{Z_\ell}^{Z_{\ell+1}} C_k^1(Z)dZ$$

where $C_k^1(Z)$ is the approximate profile expressed in linear segments, is on the average much less than the absolute errors introduced by the instrument itself. This comparison has been made on temperature profiles taken in Lake Ontario. The segments are defined by 9 temperature-depth pairs located along the profile at points where the vertical temperature gradient changes abruptly (break points). An example is given in Figure C1.

The data for a station can be assembled on a single 80-column IBM punched card. The format currently used by CCIW and the Great Lakes Institute, University of Toronto is given below.

Format for the Digitization of BT Data

In addition to the temperature-depth pairs, each profile should be accompanied by the following information:

1) station number
2) Date: year, month, day, hour, tenth of hour (6 minutes)
3) position: lat. and long. to 1/10 minute
4) depth of lake at station position (metres)
5) indices of temperature-depth pairs which characterize the upper and lower bounds of the thermocline. Index 1 refers to the surface temperature.
6) coded information for the description of profile types.

The information for a temperature profile can be placed on a single 80-column IBM card under the format listed in the following table:

Zone	Data	Format
1 – 3 inclusive	Station Number	I3
4 – 5 inclusive	year	I2
6 – 7 inclusive	month	I2
8 – 9 inclusive	day	I2
10 – 12 inclusive	hour	F3.1 (no sign or decimal)
13 – 14 inclusive	lat. (degrees)	I2
15 – 17 inclusive	lat. (minutes)	F3.1 (no sign or decimal)
18 – 19 inclusive	Long. (degrees)	I2
20 – 22 inclusive	Long. (minutes)	F3.1 (no sign or decimal)
23 – 25 inclusive	Surface Temp. °C to nearest 0.1°C	
26 – 73 inclusive	up to 8 temperature – * depth pairs left-filled	8 (F3.1, I3)
74 – 76 inclusive	depth to bottom (m) **	I3
77 inclusive	index of point marking top of thermocline	I1
78 inclusive	index of point marking bottom of thermocline	I1
79 – 80 inclusive	descriptive codes	

*If only 4 pairs are needed then columns 26 - 49 are used, and columns 50 - 73 remain blank.

**Temperatures are recorded to the nearest 0.1°C, depths to the nearest metre.

APPENDIX D

FORTRAN LISTING OF SPLOTCH AS CODED FOR A CDC 6400 COMPUTER

```
      PROGRAM SPLOTCH(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
      REAL NO
      LOGICAL STOP,MOVE
      DIMENSION A(3,4),Y(3)
      DIMENSION TI(8),Z(10)
      DIMENSION ZSTAT(100),ZBO(154,57)
      DIMENSION NHYP(100,2),VOL(100)
      COMMON/COEFF/AA(5),BB(5)
      COMMON/RUF/C(100,9),ZM(100,9),Y1,Y2,Y3,SCAFC
      COMMON/VARS/ISTAT(100),JSTAT(100),STOP(100)
      COMMON/VNAM/NSTAT(100)
      COMMON/VARD/PSTAT(100,2,2),DLAT
      COMMON/VCON/IM,JM,KM
      COMMON/FACT/GM,PHIM,SLAT,SLONG
      COMMON/VARG/NO(154,57)
C
C
C
C     DESCRIPTION OF INPUTS, OUTPUTS, AND CONTROL PARAMETERS
C
C
C     IMAX,JMAX            MAXIMUM DIMENSIONS OF SPATIAL ARRAYS (154,55)
C     KMM                  MAXIMUM NUMBER OF STATIONS (100)
C     Z(KP),KP=1,KDM       LEVELS DEFINING LAYER DEPTHS (METRES)
C                          KDM.LE.10
C
C     DLAT                 GRID LENGTH (KM)
C     DA                   CELL AREA   (M**2)
C     AA(L),BB(L),L=1,5    COEFFICIENTS DEFINING TRANSFORMATION BETWEEN
C                          GEOGRAPHICAL COORDINATES (LAT,LONG) AND GRID
C                          COORDINATES (I,J)
C
C     PHIM,GM              LAT AND LONG OF GRID ORIGIN (DEGREES)
C     IM,JM                DIMENSION OF GRID (SPATIAL ARRAYS)
C     ZBO(I,J) I=1,IM      DEPTH OF CELL(I,J) BELOW CHART DATUM(METRES)
C              J=1,JM
C
C     HD                   HEIGHT OF WATER LEVEL ABOVE CHART DATUM (METRES)
C     NSTAT(K),K=1,KM      STATION NUMBER   (1 TO 999)
C     PSTAT(K,1,1)         LATITUDE OF KTH STATION,DEGREES AND MINUTES
C     PSTAT(K,1,2)
C     PSTAT(K,2,1)         LONGITUDE OF KTH STATION, DEGREES AND MINUTES
C     PSTAT(K,2,2)
C
C     C(K,KL),ZM(K,KL)     CONCENTRATION-DEPTH PAIRS (DEPTHS IN METRES)
C                          MAX KL=9
C
C     SCAFC                COEFFICIENT USED TO CONVERT OUTPUT TO DESIRED
C                          UNITS
C
```

```
C
C
C          ZSTAT(K)          MAXIMUM SAMPLING DEPTH AT KTH STATION
C          CONLAP            CONTRIBUTION OF AN INDIVIDUAL STATION
C                            (WEIGHTED VALUE) TO THE VOLUME CONTENT
C                            FOR A LAYER
C          CONLAY            CONTRIBUTION OF A LAYER TO THE TOTAL
C                            VOLUME CONTENTS
C          CONTO             TOTAL VOLUME CONTENTS OF THE LAKE
C
C
C
C          INPUT DECK
C
C          CARD NO.          VARIABLES                 FORMAT
C
C          1                 TI(L),L=1,8               8A10
C                            TITLE CARD FOR DIGITAL MAP
C
C          2                 HD  WATER LEVEL           F10.2
C          3                 KDM,Z(KP),KP=1,KDM        13,10F7.2
C          4                 DLAT,PHIM,GM,IM,JM        F4.0,2F10.5,213
C          5                 AA(I)  I=1,5              5E14.6
C          6                 BB(I)  I=1,5              5E14.6
C          7                 SCAFC, TI(L) L=1,6        E12.5,8X,6A10
C                            SCALE FACTOR AND OUTPUT UNITS
C
C          8+                ZBO(I,J) I=1,IM, J=1,JM   19F4.0
C
C                            CONCENTRATION DEPTH DATA
C
C          1                 TI(L) L=1,8               8A10
C                            TITLE CARD FOR CONCENTRATION DATA
C
C          2+                NSTAT(K),((PSTAT(K,L,M),L=1,2),7=1,2),
C                            C(K,1),(C(K,N),ZM(K,N),N=2,9)
C                                                      FORMAT 224
C          BLANK CARD AT END OF STATION DATA
```

```
C
C           READ IN CONTROL DATA
C*****************************************************************************
C
      DATA IMAX,JMAX,KMM/154,57,100/
      READ(5,200) TI
      READ(5,210)HD
      READ(5,600) KDM,(Z(KP),KP=1,KDM)
      READ(5,201) DLAT,PHIM,GM,IM,JM
      DA=(DLAT*1000.)**2
      READ(5,601)  (AA(I),I=1,5),(BB(I),I=1,5)
C
      WRITE(6,607) (TI(I),I=1,8)
      WRITE(6,606) PHIM,GM
      WRITE(6,610)HD
      WRITE(6,605) (AA(I),I=1,5),(BB(I),I=1,5)
      READ(5,611)SCAFC,(TI(L),L=1,6)
      WRITE(6,612)(TI(L),L=1,6)
C
C           TEST FOR ADEQUATE PROGRAM DIMENSION
C
      IF((IM.LE.IMAX).AND.(JM.LE.JMAX))GO TO 1
      WRITE(6,202)IM,JM,IMAX,JMAX
      STOP
    1 CONTINUE
C
C           READ IN MEAN DEPTHS OF GRID CELLS
C
C
      DO 346 J=1,JM
      READ(5,203)(ZBO(I,J),I=1,152) (*)
      ZBO(153,J)=0. $ ZBO(154,J)=0.
  346 CONTINUE
      DO 347 I=1,IM
      DO 348 J=1,JM
      IF(ZBO(I,J).EQ.0.)GO TO 348
      ZBO(I,J)=ZBO(I,J)+HD
  348 CONTINUE
  347 CONTINUE
C
```

(*) In this case IM = 154. The original bathymetric data was arranged with IM = 152. Two extra rows of zero-depth cells were added for the sake of the aesthetic value of the output plots.

```
C
C          READ IN STATION DATA. DATA COMMENCES WITH TITLE CARD (FORMAT 8A10)
C          ENDS WITH BLANK CARD
C
      READ(5,200)TI
      WRITE(6,607)TI
      DO 100 K=1,KMM
      READ(5,224) NSTAT(K),((PSTAT(K,I,J),I=1,2),J=1,2)  ,C(K,1),(C(K,KL
     C),ZM(K,KL),KL=2,9)
      IF(NSTAT(K).LE.0)GOTO 2
      ZM(K,1)=0.0
      DO 191 II=1,9
      IJ=10-II
      IF(ZM(K,IJ).NE.0.)GO TO 192
  191 CONTINUE
  192 ZSTAT(K)=ZM(K,IJ)
  100 CONTINUE
C
C          FLAG.  PROGRAM SPACE FILLED BY STATION DATA
C
      WRITE(6,205)NSTAT(K)
      K=K+1
C
    2 KM=K-1
C
C          CHECK ON SAMPLING AND INTEGRATION DEPTHS. IF Z(KDM) IS GREATER
C          THAN MAX ZSTAT(K),Z(KDM) IS CHANGED TO MAX ZSTAT(K)
C
      ZSM=1.0
      DO 500 K=1,KM
      IF(ZSTAT(K).LE.ZSM)GO TO 500
      ZSM=ZSTAT(K) $ KSM=K
  500 CONTINUE
      DO 501 KD=1,KDM
      IF(Z(KD).GT.ZSM) GO TO 502
  501 CONTINUE
      GO TO 503
  502 KDM=KD
      Z(KD)=ZSM
  503 CONTINUE
C
C          CALCULATION OF INDICES ISTAT(K),JSTAT(K), OF CELL ENCLOSING
C          STATION NSTAT(K)
C
      DO 104 K=1,KM
      CALL SEED(K,10,JO)
      IF(((IO.GE.1).AND.(IO.LE.IM)).AND.((JO.GE.1).AND.(JO.LE.JM)))
     1GO TO 105
C
C          FLAG AND STOP. ERRONEOUS STATION DATA
C
      WRITE(6,206)NSTAT(K)
      STOP
C
  105 ISTAT(K)=IO $ JSTAT(K)=JO
  104 CONTINUE
C
C
```

MAIN BODY OF THE PROGRAM STARTS HERE

```
      KDMM=KDM-1
      CONTO=0.0
C
C ***************************************************************************************
C          VOLUME CONTENT OF EACH LAYER IS COMPUTED IN THIS LOOP
C
C
      DO 101  KD=1,KDMM
C
C          INITIALIZE ARRAY NO(I,J)
C
      DO 102 I=1,IM
      DO 103 J=1,JM
      IF(ZBO(I,J).GT.Z(KD))50,51
   50 NO(I,J)=0.
      GO TO 103
   51 NO(I,J)=-100.
  103 CONTINUE
  102 CONTINUE
C
C      CHOOSE STATIONS WHOSE PROFILES PASS THROUGH KD=TH LAYER
C      AND ASSIGN CENTRE CELLS IN ARRAY NO(I,J)
C
      DO 120 K=1,KM
      IF(ZSTAT(K).GE.Z(KD+1))52,53
   52 STOP(K)=.T.
      I=ISTAT(K)
      J=JSTAT(K)
      IF(NO(I,J).EQ.0.)GO TO 54
      CALL SLUFF(I,J,MOVE)
      IF(MOVE) GO TO 54
C
C          FLAG AND STOP. ERRONEOUS STATION DATA
C
      WRITE(6,250)NSTAT(K),ZSTAT(K),I,J,ZBO(I,J)
      STOP
C
   54 NO(I, J)=FLOAT(K)+0.25
      GO TO 120
   53 STOR(K)=.F.
  120 CONTINUE
      DO 121 K=1,KM
  121 IF(STOR(K))GO TO 122
      WRITE(6,613) Z(KD),Z(KD+1)
      STOP
  122 CONTINUE
```

```
C
C          THIS SEGMENT CONTAINS THE ALGORITHM WHICH GENERATES
C          CONCENTRIC CIRCLES ABOUT THE STATION POSITIONS.
C          REPEATED UNTIL NO(I,J) IS FILLED
C
        SK=1.
      3 WI=SK
        WJ=0.
        WMA=SK
        KTEST=0
      4 RA=WI*WI+WJ*WJ
        RMA=SK*SK
        RMI=(SK-1)*(SK-1)
        IF((RMI.LT.RA).AND.(RA.LE.RMA))5,6
      5 W=.1*(RMA-RA)/(2*SK-1.)
        CALL FILL(WI,WJ,W,SK,KTEST)
        IF(WJ.GE.WI)10,9
     10 IF(KTEST.EQ.0)12,11
     11 SK=SK+1.
        GO TO 3
      6 IF(RA.GT.RMA)7,8
      8 WI=WMA
        WJ=WJ+1.
        GO TO 4
      7 WMA=WMA-1.
      9 WI=WI-1.
        GO TO 4
     12 DO 107 I=1,IM
        DO 108 J=1,JM
        IF(NO(I,J).EQ.0.)GO TO 11
    108 CONTINUE
    107 CONTINUE
C
C          THIS SEGMENT COUNTS THE CELLS ASSIGNED TO STATION NSTAT(K)
C          AND COMPUTES NHYP(K,L) AND VOL(K) FOR THE LAYER
C
        DO 109 K=1,KM
        NHYP(K,1)=0
        NHYP(K,2)=0
    109 VOL(K)=0.
        DO 110 I=1,IM
        DO 111 J=1,JM
        K=INT(NO(I,J))
        ZW=ZBO(I,J)
        IF(K.LE.0)GO TO 111
        NHYP(K,1)=NHYP(K,1)+1
        IF(ZW.LT.Z(KD+1))GO TO 15
        NHYP(K,2)=NHYP(K,2)+1
        VOL(K)=VOL(K)+Z(KD+1)-Z(KD)
        GO TO 111
     15 VOL(K)=VOL(K)+ZW-Z(KD)
    111 CONTINUE
    110 CONTINUE
```

```
C
C          THIS SEGMENT CALCULATES Y1,Y2, AND Y3 FROM NHYP(K,L) AND
C          VOL(K) AND THEN PROCEEDS TO INTEGRATE STATION PROFILE
C          DATA THROUGH THE LAYER.
C
      WRITE(6,207)Z(KD),Z(KD+1)
C
      CONLAY=0.0
C
      DO 112 K=1,KM
      IF(.NOT.STOR(K))GO TO 112
      A(1,1)=1.
      A(2,1)=1.
      A(3,1)=Z(KD+1)-Z(KD)
      A(1,2)=Z(KD)
      A(2,2)=Z(KD+1)
      A(3,2)=(Z(KD+1)**2-Z(KD)**2)/2.
      A(1,3)=Z(KD)**2
      A(2,3)=Z(KD+1)**2
      A(3,3)=(Z(KD+1)**3-Z(KD)**3)/3.
      A(1,4)=NHYP(K,1)
      A(2,4)=NHYP(K,2)
      A(3,4)=VOL(K)
      CALL RELIN(3,4,1,A,Y,ITST)
      IF(ITST.EQ.0)GO TO 16
C
C          FLAG AND STOP. CANNOT FIND A REAL SOLUTION FOR Y1,Y2, Y3
C
      WRITE(6,208)NSTAT(K)
      STOP
C
   16 Y1=Y(1)*DA
      Y2=Y(2)*DA
      Y3=Y(3)*DA
C
      CONLAP=0.0
C
      CALL LAYIN(K,Z(KD),Z(KD+1),CONLAP)
      WRITE(6,209)NSTAT(K), Z(KD), Z(KD+1),Y1,Y2,Y3,CONLAP
      CONLAY =CONLAY + CONLAP
C
  112 CONTINUE
C
      WRITE(6,759) CONLAY
      CONTO= CONTO + CONLAY
C
      CALL SPLOT(Z(KD),Z(KD+1))
  101 CONTINUE
C
C ***************************************************************************************
C
C          END OF CALCULATION. FINAL RESULTS OUTPUT
C
      WRITE(6,761) CONTO
C
```

```
C
C          COMPUTATION OF THE VOLUME OF THE LAKE BELOW THE DEEPEST
C          LEVEL FOR WHICH THERE IS CONCENTRATION DATA.
C          ESTIMATE OF VOLUME CONTENT OF THIS LAYER
C
C
      VOLB=0.
      DO 800 J=1,JM
      DO 800 I=1,IM
      IF(ZBO(I,J).LE.ZSM)GO TO 800
      VOLB=VOLB+ZBO(I,J)-ZSM
  800 CONTINUE
      VOLB=VOLB*DA
      DO 802 L=1,9
      LL=10-L
      IF(ZM(KSM,LL).GT.0.)GO TO 803
  802 CONTINUE
  803 TB=C(KSM,LL) $ HVOLB=SCAFC*VOLB*TB
      CONTOT=CONTO+HVOLB
      WRITE(6,763)VOLB,ZSM,NSTAT(KSM),TB,HVOLB,CONTOT
 4055 STOP
C
C          FORMATS
C
  200 FORMAT(8A10)
  201 FORMAT(F4.0,2F10.5,213)
  202 FORMAT(5X,52HPROGRAM-DIMENSION-INSUFFICIENT-TO-RECEIVE DIGITAL
     1MAP;/,5X,3HIM=,I4,5X,3HJM=,I4,5X,5HIMAX=,I4,5X,5HJMAX=,14)
  203 FORMAT(19F4.0)
  205 FORMAT(5X,37HPROGRAM SPACE FILLED BY STATION DATA.,/,
     120HLAST STATION NUMBER=,I4)
  206 FORMAT(5X,7HSTATION,I4,2X,23HLIES OUTSIDE THE REGION)
  207 FORMAT(1H1,6X,39HHYPSOMETRIC CONSTANTS FOR LAYER BETWEEN,F6.1,1X,
     110HMETRES AND,F6.1,1X,7HMETRES.,//,6X,46HA(Z)M**2=Y1+Y2*Z+Y3*Z**2
     2WHERE Z IS IN METRES.,////,1X,
     33HSTA,7Z,2HZU,8Z,2HZL,10X,2HY1,13X,2HY2,13X,2HY3,18X,14HVOLUME CON
     4TENT)
  208 FORMAT(/,6X,28HNO SOLUTION FOR Y AT STATION,I4)
  209 FORMAT(1X,I2,2(4X,F6.1),9X,3(E10.3,5X),E14.5)
  210 FORMAT(F10.2)
  224 FORMAT(I3,9X,2(F2.0,F3.1),F3.1,8(F3.1,,F3.0))
  250 FORMAT(6X,7HSTATION,I4,22HWITH MAX SAMPLE DEPTH=,F8.1,1HM,2X,
     B18HASSIGNED TO CELL (,I4,1H,,I4,11H) OF DEPTH=,F8.1,1HM)
  600 FORMAT(I3,10F7.2)
  601 FORMAT(5E14.6)
  605 FORMAT(1HO,* TRANSFORMATION COEFFICIENTS-X- *5E15.4  /  1X ,
     C          *                                  -y- *5E15.4 )
  606 FORMAT(1HO,* MAP ORIGIN (IN DEGREES) IS (*F5.2,1H,,F5.2,1H))
  607 FORMAT(1H1,8A10 /// )
  610 FORMAT(/,5X,*WATER LEVEL IS*,F8.2,2X,*METRES ABOVE CHART DATUM*,/)
  611 FORMAT(E12.5,8Z,6A10)
  612 FORMAT(/,5X,6A10,/)
  613 FORMAT(5X,*WIERD EVENT - NO STATIONS SELECTED *,2F10.2)
  759 FORMAT(1X,/// 40X,* VOLUME CONTENT OF THIS LAYER IS*,4X,E15.5)
  761 FORMAT(1HO,/// 40X,* TOTAL VOLUME CONTENT OF LAKE IS*,4X,E15.5)
  763 FORMAT(//,5X,*VOLUME OF LAKE BELOW DEEPEST SAMPLING DEPTH =*,
     1E12.5,*  M**3*,//,5X,*DEEPEST OBSERVATION AT*,F5.1,*M.  OCCURS AT
     2STATION*,I4,*   VALUE = *,F5.1,*DEG.C*,//,5X,*CONTRIBUTION TO VOLU
     3ME CONTENTS = *,E12.5,*  ADJUSTED TOTAL CONTENTS = *,E12.5)
C
      END
```

```
      SUBROUTINE LAYIN(K,ZU,ZL,CONLAP)
C          PERFORMS INTEGRATION OF CONCENTRATION PROFILE OF STATION NSTAT(K) OVER
C          THE VERTICAL DISTANCE ZU ≤ Z ≤ ZL USING THE HYPSOMETRIC CURVE FOR THE
C          VOLUME OF INFLUENCE OF STATION NSTAT(K)
C
      DIMENSION CW(10),ZW(10)
      COMMON/RUF/C(100,9),ZM(100,9),Y1,Y2,Y3,SCAFC
      COMMON/VNAM/NSTAT(100)
      SIG(A,A1,B,B1,Z)=A+((A1-A)/(B1-B))*(Z-B)
C          (THIS IS THE FUNCTION USED FOR INTERPOLATION)
      JM= 9
      DO 100 J=1,JM
      IF(ZM(K,J).GT.ZU) GO TO 1
  100 CONTINUE
C          ERROR EXIT
      WRITE(6,2) NSTAT(K)
      WRITE(6,3)(C(K,J),ZM(K,J),J=1,JM)
      CONLAP=0.0
      STOP
    1 IF((J-1).GT.0)GO TO 4
      JJ=J+1
      CW(1) =SIG(C(K,J),C(K,JJ),ZM(K,J),ZM(K,JJ),ZU)
      GO TO 5
    4 JJ=J-1
      CW(1)=SIG(C(K,JJ),C(K,J),ZM(K,JJ),ZM(K,J),ZU)
    5 ZW(1)=ZU
      IW=1
      DO 101 JW=J,JM
      IW=IW+1
      IF(ZM(K,JW).GE.ZL) GO TO 7
      CW(IW)=C(K,JW)
      ZW(IW)=ZM(K,JW)
  101 CONTINUE
C          ERROR EXIT
      WRITE(6,9) NSTAT(K)
      WRITE(6,3)(C(K,J),ZM(K,J),J=1,JM  )
      STOP
    7 JJ=JW-1
      CW(IW)=SIG(C(K,JJ),C(K,JW),ZM(K,JJ),ZM(K,JW),ZL)
      ZW(IW)=ZL
      IWM=IW
      CONLAP=0.0
      DO 102 IW=2,IWM
      IPAIN=IW-1
      IF(ZW(IW).EQ.ZW(IPAIN)) ZW(IW)=ZW(IW)+0.1
      AI=(CW(IW)-CW(IW-1))/(ZW(IW)-ZW(IW-1))
      AO=CW(IW-1)-AI*ZW(IW-1)
      H1=AO*Y1*(ZW(IW)-ZW(IW-1))
      H2=(AO*Y2+AI*Y1)*(ZW(IW)**2-ZW(IW-1)**2)/2.0
      H3=(AO*Y3    +AI*Y2    )*(ZW(IW)**3-ZW(IW-1)**3)/3.0
      H4=AI*Y3    *(ZW(IW)**4-ZW(IW-1)**4)/4.0
  102 CONLAP= CONLAP +(H1+H2+H3+H4)*SCAFC
      RETURN
    2 FORMAT(1HO,*PROFILE DOES NOT EXTEND TO TOP LAYER*/8HSTATION ,I4)
    3 FORMAT(1X,5(3X,2HC=,F5.2,3X,3HZM=,F5.2,*/*))
    9 FORMAT(1HO,*PROFILE DOES NOT EXTEND TO BOTTOM OF LAYER*/
     A8HSTATION,I4)
      END
```

```
      SUBROUTINE SLUFF(I,J,MOVE)
C
C         SLUFF IS CALLED IF THE SAMPLING DEPTH IS GREATER THAN THE
C         MAP DEPTH. THE ADJOINING CELLS ARE SEARCHED AND IF ONE OF
C         THEM IS DEEP ENOUGH,MOVE IS ASSIGNED VALUE .T. WHICH ALLOWS
C         THE COMPUTATION TO CONTINUE. OTHERWISE MOVE IS .F. AND
C         THE MAIN PROGRAM FLAGS AN ERROR AND STOPS.
C
      LOGICAL MOVE
      DIMENSION II(8),JJ(8)
      COMMON/VARG/NO(154,57)
      DATA(II(K),K=1,8)/0,1,1,1,0,-1,-1,-1/
      DATA(JJ(K),K=1,8)/1,1,0,-1,-1,-1,0,1/
      MOVE=.T.
      DO 100 K=1,8
      IO=I+II(K) $ JO=J+JJ(K)
      IF(NO(IO,JO).EQ.0.) RETURN
  100 CONTINUE
      MOVE=.F.
      RETURN
      END
```

```
      SUBROUTINE SEED(K,IO,JO)
C
C         SEED COMPUTES THE INDICES,IO,JO, OF THE GRID CELL
C         WHICH CONTAINS THE POINT SPECIFIED BY THE GEOGRAPHICAL
C         COORDINATES OF STATION NSTAT(K).
C
C
      REAL LAT,LONG
      COMMON/COEFF/A(5),B(5)
      COMMON/FACT/GM,PHIM,LAT,LONG
      COMMON/VARD/PSTAT(100,2,2),DLAT
      COMMON/VNAM/NSTAT(100)
      LAT=PSTAT(K,1,1)+PSTAT(K,2,1)/60.0
      LONG=PSTAT(K,1,2) +PSTAT(K,2,2)/60.0
      G=GM-LONG
      P=LAT-PHIM
      X=G*A(1)+P*A(2)+P*G*A(3)+(G**2)*A(4)+(P**2)*A(5)
      Y=G*B(1)+P*B(2)+P*G*B(3)+(G**2)*B(4)+(P**2)*B(5)
      IO=(X/DLAT)+1
      JO=(Y/DLAT)+1
      RETURN
      END
```

```
      SUBROUTINE SPLOT(ZU,ZL)
      REAL NO
      COMMON/VARG/NO(154,57)
      COMMON/VNAM/NSTAT(100)
      COMMON/VCON/IM,JM,KM
C
C     THIS SUBROUTINE DRAWS A MAP OF THE WEIGHTING AREAS
C
      IF(JM.GT.67) GO TO 10
      WRITE(6,203)ZU,ZL
      DO 100 I=1,IM
      DO 101 J=1,JM
      K=INT(NO(I,J))
      IF(K.LE.0) GO TO 101
      NO(I,J)=FLOAT(NSTAT(K))
  101 CONTINUE
  100 CONTINUE
      DO 102 I=1,IM
      WRITE(6,201) (NO(I,J),J=1,JM)
  102 CONTINUE
      RETURN
   10 WRITE(6,202)JM
      RETURN
  201 FORMAT(1X,67F2.0)
  202 FORMAT(5X,3HJM=,I4,15HNOT ENOUGH ROOM)
  203 FORMAT(1HO,5X,40HMAP OF WEIGHTING AREAS FOR LAYER BETWEEN,F8.1,
     15HM AND,F8.1,1HM,////)
      END
```

```
      SUBROUTINE FILL(WI,WJ,W,SK,KTEST)
C
C          FILL ASSIGNS GRID CELLS TO INDIVIDUAL STATIONS. IT CONTAINS
C          THE LOGIC WHICH DECIDES ON A STATION≠S CLAIM TO A CELL
C          KTEST=1 AT  EXIT  IF A PREVIOUSLY EMPTY CELL HAS BEEN
C          FILLED, 0 OTHERWISE
C
C
      REAL NO
      LOGICAL STOR
      DIMENSION IWR(8),JWR(8)
      COMMON/VARG/NO(154,57)
      COMMON/VARS/ISTAT(100),JSTAT(100),STOR(100)
      COMMON/VCON/IM,JM,KM
      KTEST=0
      IW=INT(WI)
      JW=INT(WJ)
      IWR(1)=IW
      IWR(2)=-IW
      IWR(3)=IW
      IWR(4)=-IW
      IWR(5)=JW
      IWR(6)=-JW
      IWR(7)=JW
      IWR(8)=-JW
      JWR(1)=JW
      JWR(2)=JW
      JWR(3)=-JW
      JWR(4)=-JW
      JWR(5)=IW
      JWR(6)=IW
      JWR(7)=-IW
      JWR(8)=-IW
      DO 100 KR=1,8
      DO 101 K=1,KM
      IF(.NOT.STOR(K))GO TO 101
      I=ISTAT(K)+IWR(KR)
      J=JSTAT(K)+JWR(KR)
      IF(((I.GT.IM).OR.(I.LT.1)).OR.((J.GT.JM).OR.(J.LT.1)))GO TO 101
      IF(NO(I,J))101,102,103
  102 NO(I,J)=FLOAT(K)+W
      KTEST=1
      GO TO 101
  103 WO=AMOD(NO(I,J),1.)
      KS=INT(NO(I,J))
      IRA=(I-ISTAT(KS))**2+(J-JSTAT(KS))**2
      IF(IRA.LE.(INT(SK)-1)**2)GO TO 101
      IF(W.GT.WO)GO TO 102
  101 CONTINUE
  100 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE RELIN(N,N1,N2,A,X,IT)
C
C
C          RELIN SOLVES AN N BY N SET OF LINEAR EQUATIONS
C          WITH N2 RIGHT HAND MEMBERS. GAUSS≠S METHOD IS USED.
C          FORM OF EQUATIONS≤
C          A(I,1)*X(1,J)+A(I,2)*X(2,J)+...+A(I,N)*X(N,J)=A(I,N+J)
C          I=1,2,...,N
C          J=1,2,...,N2
C          N1=N+N2
C          ARRAY X(N,N2) CONTAINS N2 SOLUTION VECTORS AT END OF
C          COMPUTATION
C          IF THE MATRIX A(I,K), I=1,N,K=1,N IS SINGULAR
C          CONTROL IS RETURNED TO MAIN PROGRAM AND IT HAS
C          VALUE 1.  IT=0 SIGNIFIES NONSINGULAR MATRIX.
C
C
      DIMENSION A(N,N1),X(N,N2)
      IT=0
      ZERO=1.0E-15
      DO40I=1,N
      JM=0
   42 JM=JM+1
      IF(JM.GT.N)IT=1
      IF(IT.EQ.1)RETURN
      IF(A(I,JM).EQ.0.)GOTO42
      AMX=ABS(A(I,JM))
      DO41J=JM,N
   41 IF(ABS(A(I,J)).GT.AMX)AMX=ABS(A(I,J))
      DO43J=1,N1
   43 A(I,J)=A(I,J)/AMX
   40 CONTINUE
      K=1
   22 M=1
      IS=K+1
      IF(IS.GT.N)IS=MOD(IS,N)
      IF(ABS(A(K,K)).GT.ZERO)GOTO20
   25 IF(ABS(A(IS,K)).GT.ZERO)GOTO28
      IS=IS+1
      IF(IS.GT.N)IS=MOD(IS,N)
      M=M+1
      IF(M.GT.N)IT=1
      IF(IT.EQ.1)RETURN
      GO TO 25
   20 K=K+1
      IF(K.LE.N)GOTO22
      GOTO24
   28 DO29J=1,N1
   29 A(K,J)=A(K,J)+A(IS,J)
      GOTO20
   24 CONTINUE
```

```
      N3=N-1
      DO8KEEP=1,N3
      DO1I=KEEP,N
      S=A(I,KEEP)
      IF(ABS(S).LE.ZERO)GOTO1
      DO2J=1,N1
    2 A(I,J)=A(I,J)/S
    1 CONTINUE
      K2=KEEP+1
      DO5I=K2,N
      IF(ABS(A(I,KEEP)).LE.ZERO)GOTO5
      DO6J=1,N1
    6 A(I,J)=A(I,J)-A(KEEP,J)
    5 CONTINUE
      IS=K2+1
      IF(ABS(A(K2,K2)).GT.ZERO)GOTO8
   55 IF(IS.GT.N)IT=1
      IF(IT.EQ.1)RETURN
      IF(ABS(A(IS,K2)).GT.ZERO)GOTO58
      IS=IS+1
      GOTO55
   58 DO59J=1,N1
   59 A(K2,J)=A(K2,J)+A(IS,J)
    8 CONTINUE
      A(N,N1)=A(N,N1)/A(N,N)
      A(N,N)=1.
      DO14K=1,N
   14 IF(ABS(A(K,K)).LE.ZERO)IT=1
      IF(IT.EQ.1)RETURN
      DO12K=1,N2
      DO9I=1,N
    9 X(I,K)=0.
      DO10I=1,N
      L=N+1-I
      SL=0.
      DO11J=L,N
   11 SL=SL+A(L,J)*X(J,K)
   10 X(L,K)=A(L,N+K)-SL
   12 CONTINUE
      RETURN
      END
```

OUTPUT FROM PROGRAM

At the beginning, the program outputs the information relative to the digital map used, the lake level assumed, the output units, and the information on the title card preceding the profile data.

The output for each layer consists of a title giving the layer depths and a table listing the station numbers appearing in the calculation, the constants Y_1, Y_2, Y_3 used to define $A_{\ell_k}(Z)$ (Fig. 1) and the contribution of each station in the list to the volume content of the layer. The sum of these contributions appears at the bottom of the table.

A two-dimensional map of the horizontal distribution of areas of influence is constructed by the subroutine SPLØT (called after statement 112 in the main program). Each cell is represented by a zone two characters wide and one line deep. The station number assigned to each cell is printed in the zone; if the cell depth is too shallow for the layer considered, asterisks are used to fill the space on the plot. The plot serves only to verify the functioning of the program and is usually suppressed for operational runs.

A final stage in the computation and output is an estimate of the volume content of the lake below the depth of the deepest sample. Here it is assumed that the concentration measured at the deepest sampling point applies to all deeper depths.